

# NAG Fortran Library Routine Document

## G05MCF

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

### 1 Purpose

G05MCF generates a vector of pseudo-random integers from the discrete negative binomial distribution with parameter  $m$  and probability  $p$  of success at a trial.

### 2 Specification

```
SUBROUTINE G05MCF(MODE, M, P, N, X, IGEN, ISEED, R, NR, IFAIL)
INTEGER          MODE, M, N, X(N), IGEN, ISEED(4), NR, IFAIL
real           P, R(NR)
```

### 3 Description

G05MCF generates  $n$  integers  $x_i$  from a discrete negative binomial distribution, where the probability of  $x_i = I$  ( $I$  successes before  $m$  failures) is

$$P(x_i = I) = \frac{(m + I - 1)!}{I!(m - 1)!} \times p^I \times (1 - p)^m, \quad I = 0, 1, \dots$$

The variates can be generated with or without using a search table and index. If a search table is used then it is stored with the index in a reference vector and subsequent calls to G05MCF with the same parameter value can then use this reference vector to generate further variates.

One of the initialisation routines G05KBF (for a repeatable sequence if computed sequentially) or G05KCF (for a non-repeatable sequence) must be called prior to the first call to G05MCF.

### 4 References

Knuth D E (1981) *The Art of Computer Programming (Volume 2)* (2nd Edition) Addison-Wesley

### 5 Parameters

1: MODE – INTEGER *Input*

*On entry:* a code for selecting the operation to be performed by the routine:

MODE = 0

Set up reference vector only.

MODE = 1

Generate variates using reference vector set up in a prior call to G05MCF.

MODE = 2

Set up reference vector and generate variates.

MODE = 3

Generate variates without using the reference vector.

*Constraint:*  $0 \leq \text{MODE} \leq 3$ .

- 2: M – INTEGER *Input*  
*On entry:* the number of failures,  $m$ , of the distribution.  
*Constraint:*  $M \geq 0$ .
- 3: P – *real* *Input*  
*On entry:* the parameter  $p$  of the negative binomial distribution representing the probability of success at a single trial.  
*Constraint:*  $0.0 \leq P < 1.0$ .
- 4: N – INTEGER *Input*  
*On entry:* the number,  $n$ , of pseudo-random numbers to be generated.  
*Constraint:*  $N \geq 1$ .
- 5: X(N) – INTEGER array *Output*  
*On exit:* the  $n$  pseudo-random numbers from the specified negative binomial distribution.
- 6: IGEN – INTEGER *Input*  
*On entry:* must contain the identification number for the generator to be used to return a pseudo-random number and should remain unchanged following initialisation by a prior call to one of the routines G05KBF or G05KCF.
- 7: ISEED(4) – INTEGER array *Input/Output*  
*On entry:* contains values which define the current state of the selected generator.  
*On exit:* contains updated values defining the new state of the selected generator.
- 8: R(NR) – *real* array *Input/Output*  
*On exit:* the reference vector.
- 9: NR – INTEGER *Input*  
*On entry:* the dimension of the array R as declared in the (sub)program from which G05MCF is called.  
*Suggested value:*  $NR = 20 + (20\sqrt{N \times P} + 30 \times P)/(1 - P)$  approximately.  
*Constraints:*  
if MODE = 0 or 2, then
$$NR > \text{int}\left(\frac{M \times P + 7.15\sqrt{M \times P} + 20.15 \times P}{1 - P} + 8.5\right) - \max\left(0, \text{int}\left(\frac{M \times P - 7.15\sqrt{M \times P}}{1 - P}\right)\right) + 7;$$
if MODE = 1, then NR should remain unchanged from the previous call to G05MCF;  
if MODE = 3, then R is not referenced.
- 10: IFAIL – INTEGER *Input/Output*  
*On entry:* IFAIL must be set to 0, -1 or 1. Users who are unfamiliar with this parameter should refer to Chapter P01 for details.  
*On exit:* IFAIL = 0 unless the routine detects an error (see Section 6).  
For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the

value 1 is recommended. Otherwise, for users not familiar with this parameter the recommended value is 0. **When the value  $-1$  or  $1$  is used it is essential to test the value of IFAIL on exit.**

## 6 Error Indicators and Warnings

If on entry IFAIL = 0 or  $-1$ , explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry,  $N < 1$ .

IFAIL = 2

On entry, NR is too small when MODE = 0 or 2 (see Section 5).

IFAIL = 3

On entry,  $P < 0.0$   
or  $P \geq 1.0$ .

IFAIL = 4

On entry,  $M < 0$ .

IFAIL = 5

On entry, MODE  $< 0$   
or MODE  $> 3$ .

IFAIL = 6

MODE = 0 or 2 and P is so close to 1 that NR would have to be larger than the largest representable integer. Use MODE = 3 in this case.

IFAIL = 7

P or M is not the same as when R was set up in a previous call with MODE = 0 or 2.

## 7 Accuracy

Not applicable.

## 8 Further Comments

None.

## 9 Example

The example program prints five pseudo-random integers from a negative binomial distribution with parameters  $m = 60$  and  $p = 0.999$ , generated by a single call to G05MCF, after initialisation by G05KBF.

### 9.1 Program Text

**Note:** the listing of the example program presented below uses *bold italicised* terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*      G05MCF Example Program Text
*      Mark 20 Release. NAG Copyright 2001.
*      .. Parameters ..
```

```

      INTEGER          NOUT, N, NR
      PARAMETER       (NOUT=6,N=20,NR=1)
*   .. Local Scalars ..
      real            P
      INTEGER          I, IFAIL, IGEN, M
*   .. Local Arrays ..
      real            R(NR)
      INTEGER          ISEED(4), X(N)
*   .. External Subroutines ..
      EXTERNAL         G05KBF, G05MCF
*   .. Executable Statements ..
      WRITE (NOUT,*) 'G05MCF Example Program Results'
      WRITE (NOUT,*)
*   Set the distribution parameter P
      P = 0.999e0
      M = 60
*   Initialise the seed to a repeatable sequence
      ISEED(1) = 1762543
      ISEED(2) = 9324783
      ISEED(3) = 42344
      ISEED(4) = 742355
*   IGEN identifies the stream.
      IGEN = 1
      CALL G05KBF(IGEN,ISEED)
*   Choose MODE = 3 because P close to 1
      IFAIL = 0
      CALL G05MCF(3,M,P,N,X,IGEN,ISEED,R,NR,IFAIL)
*
      WRITE (NOUT,99999) (X(I),I=1,N)
      STOP
*
99999 FORMAT (1X,I12)
      END

```

## 9.2 Program Data

None.

## 9.3 Program Results

G05MCF Example Program Results

```

49822
73300
57798
64791
73197
58394
61279
57579
56346
70705
59782
58087
61213
81333
47030
57082
67922
56991
46611
62691

```